
Zusammenfassung

Um einen zu SAS 9.2 in der Ausgabe von Grafiken hinsichtlich ihrer Editierbarkeit ver-
gleichbaren Stand zu erzielen, ist in SAS 9.3 und SAS 9.4 eine einmalige von der SAS-
Hotline mitgeteilte Anpassung der SAS Registry notwendig.
Daneben ist es vorteilhaft, die je nach Form der Abspeicherung als EMF-Grafik unter-
schiedlichen EMF-Formate zu betrachten und ihre Eigenheiten beim Editieren zu kennen.

Schlüsselwörter: EMF-Format, SAS 9.3, SAS 9.4, Grafik editieren

1 Notwendige Änderungen in der SAS Registry
Um einen zu SAS 9.2 in der Ausgabe von Grafiken hinsichtlich ihrer Editierbarkeit ver-
gleichbaren Stand zu erzielen, bedarf es unter SAS 9.3 bzw. SAS 9.4 einer einmaligen 
von der SAS-Hotline mitgeteilten Anpassung der SAS Registry.
Benötigt wird ein vorhandener oder anzulegender temporärer Ordner. Beispielhaft sei er 
C:\TEMP. In SAS 9.3 bzw. SAS 9.4 wird diese Anpassungsdatei mit Hilfe folgender 
Programmzeilen angelegt:

data _null_;
file 'C:\TEMP\emf94.sasxreg';
put '[CORE\PRINTING\PRINTERS\EMF\ADVANCED]';
put '"Description"="Enhanced Metafile Format"';
put '"Metafile Type"="EMF"';
put '"Vector Alpha"=int:0';
put '"Image 32"=int:1';

run;

Die notwendigen Änderungen in der SAS Registry werden dauerhaft durch die Zeile
proc registry import='C:\TEMP\emf94.sasxreg'; run;

ausgeführt.

Das ist schon alles.

2 Unterschiedliche EMF-Formate
Etliche Software, z.B. auch JMP, liefert editierfähige Grafiken. Das kann auch für SAS-
Anwender von Interesse sein, zumal es noch bis SAS 9.2 so war. Dann muss einmalig 
obiger Patch für Grafiken vom EMF-Format eingesetzt werden. SAS stellt andererseits 
genügend Hilfsmittel bereit, damit auf ein nachträgliches Editieren verzichtet werden 
kann. Wenn die Möglichkeit eines nachträglichen Editierens infrage kommt, ist im All-
gemeinen darauf zu achten, dass beim Editieren etliche Text-Eigenschaften wie verti-
kale oder schräge Ausrichtung verloren gehen können.
Die EMF-Formate können je nach Form der Abspeicherung unterschiedlich sein.

2.1 EMF-Grafik aus dem Grafik-Fenster heraus anlegen

2.1.1 Beispiel
goptions ftext='Arial Narrow' htext=0.6;
symbol v=dot c=blue h=0.4;
proc boxplot data=sashelp.cars;

plot Horsepower*Make / cboxes=black cboxfill=empty boxwidth=4 ;
run; quit;

2.1.2 Abspeichern als EMF-Grafik

Es werden 2 Grafiken angelegt: Panel 1 und Panel 2. Jede der beiden Grafiken kann 
vom Grafikfenster aus mittels der rechten Maustaste 

als EMF-Grafik gespeichert werden.
Um sie näher betrachten zu können, werden die Grafiken mit PS1.emf und PS2.emf be-
nannt.

2.1.3 Einfügen dieser Grafik in eine MS Office-Anwendung

Bereits beim Einfügen einer der beiden Grafiken in beispielsweise MS-WORD fällt auf, 
dass sie unvollständig zu sein scheint (Abb. 1).

Abbildung 1: PS1.emf

2.1.4 Editieren dieser Grafik

Um diese Grafik zu editieren wird bekanntlich die rechte Maustaste genutzt:
Bild bearbeiten

Die danach erscheinende Frage, ob die Grafik zu einem Microsoft Office-Zeichnungs-
objekt umgewandelt werden soll, ist zu bejahen.
Als nächster Schritt sind die über die grafischen Elemente liegenden, die gesamte Bild-
fläche einnehmenden transparenten Objekte zu löschen. Danach wir die ganze Bildflä-
che markiert. Nun zeigt sich, dass alle grafischen Elemente vorhanden sind; ihnen nur 
die Linienfarbe fehlt (Abb. 2).

Abbildung 2: PS1.emf, grafische Elemente markiert

Die fehlende Linienfarbe wird über
n

ausgewählt.

Wenn die gesamte Bildfläche markiert wurde, werden auch die Achsenbeschriftungen 
mit einer Umrandung versehen.

Für den Fall, dass eine EMF-Grafik aus dem Grafik-Fenster heraus angelegt wurde, ist 
die beschriebene Vorgehensweise weniger ein Editieren als ein notwendiges Korrigie-
ren.

2.2 EMF-Grafik mittels filename und goptions anlegen

2.2.1 Beispiel
filename f 'F:\KSFE\PS3.emf';
goptions gsfname=f gsfmode=replace device=emf 

ftext='Arial Narrow' htext=0.6;
symbol v=dot c=blue h=0.4;
proc boxplot data=sashelp.cars (WHERE=(Make<'Lexus')) ;

plot Horsepower*Make 
/ cboxes=black cboxfill=empty boxwidth=4;

run;

2.2.2 Abspeichern als EMF-Grafik

Beiden Grafiken müssen beispielsweise über den Weg der WHERE-Klausel eigene 
Dateinamen zugeordnet werden. Im Beispiel ist das die Datei PS3.emf. Die grafischen 
Optionen goptions umfassen notwendiger Weise auch die Verbindung zum Namen der 
Grafik und zum Grafik-Format.

2.2.3 Editieren dieser Grafik in einer MS Office-Anwendung 

Das Einfügen der Grafik PS3.emf läuft wie oben beschrieben ab. Sie erscheint vollstän-
dig. Das ‘Bild bearbeiten‘ zeigt, dass in diesem Fall keine über den grafischen Elemen-
ten liegenden, die gesamte Bildfläche einnehmenden transparenten Objekte zu löschen 
sind. Die Bezeichnung der Ordinatenachse wird aber um 90° gedreht. Das lässt sich ggf. 
ändern.

2.3 EMF-Grafik unter Verwendung der ODS GRAPHICS-Klammer
anlegen

2.3.1 Beispiel
ODS GRAPHICS on/ reset=all imagefmt=emf;
proc sgplot data=sashelp.cars (WHERE=(Make<'Lexus')) ;

vbox Horsepower 
/ category=Make extreme 

meanattrs=(symbol=CircleFilled size=4 color=blue)
fillattrs=(color=white);

xaxis LABELATTRS=(Family="Arial Narrow" Size=10)
VALUEATTRS=(Family="Arial Narrow" Size=7);
yaxis LABELATTRS=(Family="Arial Narrow" Size=10);

run;
ODS GRAPHICS off;



Tipps & Tricks

239

2.3.2 Abspeichern als EMF-Grafik

Mit dem Beispiel 2.3.1 wollen wir mit PROC SGPLOT eine ähnliche Grafik wie 
PS3.emf mit PROC BOXPLOT erzeugen. Da die grafischen Optionen goptions nicht 
weiter helfen, aber ebenso eine editierbare EMF-Grafik erzeugt werden soll, ist die ODS 
GRAPHICS-Option imagefmt=emf notwendig. Die Grafik wird im aktuellen Arbeits-
verzeichnis C:\Users\<Nutzer> abgelegt.

2.3.3 Editieren dieser Grafik in einer MS Office-Anwendung 

Das Einfügen der Grafik SGPlot.emf bedarf keiner Erläuterungen, ebenso der Weg zum 
‘Bild bearbeiten‘. Auch in diesem Fall sind keine über den grafischen Elementen lie-
genden, die gesamte Bildfläche einnehmenden transparenten Objekte zu löschen. Das 
Aber ist, dass die Bezeichnung der Ordinatenachse verkleinert und um 90° gedreht wird 
und die Beschriftungen der Abszissenachse nicht schräg sondern horizontal angeordnet

Abbildung 3: SGPlot.emf

und ebenfalls verkleinert werden (Abb. 4). Mindestens die Beschriftung der Ordinaten-
achse muss geändert werden.

Abbildung 4: SGPlot.emf, nach ‘Bild bearbeiten‘

2.4 EMF-Grafik mittels ODS RTF- und ODS GRAPHICS-Klammer
anlegen

2.4.1 Beispiel
options nocenter nodate nonumber ps=3000 formdlim=' '; title;
ODS RTF file='F:/KSFE/Ausgabe.rtf';
goptions reset=all device=emf

ftext='Arial Narrow' htext=0.6;
symbol v=dot c=blue h=0.4;
proc boxplot data=sashelp.cars  ;

plot Horsepower*Make 
/ cboxes=black cboxfill=empty boxwidth=4;

run; quit;

ODS GRAPHICS on / reset=all imagefmt=emf  ;
proc sgplot data=sashelp.cars  ;

vbox Horsepower 
/ category=Make extreme 

meanattrs=(symbol=CircleFilled size=4 color=blue)
fillattrs=(color=white);

xaxis LABELATTRS=(Family="Arial Narrow" Size=10)
VALUEATTRS=(Family="Arial Narrow" Size=7);
yaxis LABELATTRS=(Family="Arial Narrow" Size=10);

run;
ODS GRAPHICS off;
ODS RTF close;
ODS listing;

2.4.2 Bemerkungen zu den ODS-Klammern und den grafischen 
Optionen

Die äußere ODS-Klammer, die RTF-Klammer, wird mit einer Dateiangabe versehen, 
die für unser Beispiel nur Grafiken enthält. Die grafischen Optionen goptions, die sich 
auf PROC BOXPLOT beziehen sollen, haben und brauchen keine Verbindung zu einem 
Grafik-Namen. Die Zuweisung des Grafik-Formats device=emf kann sowohl in 
SAS 9.3 als auch in SAS 9.4 entfallen.
Die für das Grafik-Format notwendige Option imagefmt=emf in der ODS GRAPHICS-
Klammer um PROC SGPLOT muss in SAS 9.3 geschrieben werden, kann aber in 
SAS 9.4 entfallen.

Bei Beibehaltung der jeweiligen SAS-Grundeinstellung wird in SAS 9.3 nach dem 
Schließen der ODS RTF-Klammer gefragt, ob die RTF-Datei (mit MS WORD) geöffnet 
werden soll, während sie in SAS 9.4 ohne Nachfragen geöffnet wird.

2.4.3 Die EMF-Grafiken in MS WORD 

Die in MS-WORD geöffnete RTF-Datei enthält insgesamt drei Grafiken. Aus dem Teil-
programm zu PROC BOXPLOT resultieren zwei Grafiken, weil im Beispiel keine Ein-
schränkung bezüglich der Automarken z.B. durch eine WHERE-Klausel vorgenommen 
wurde. Eine Grafik wird von PROC SGPLOT erzeugt; auch hier wird im obigen Bei-
spiel die Automarke nicht eingegrenzt.
Alle drei Grafiken sind verlustfrei skalierbar und editierbar. Sollte ein Bearbeiten der 
Grafiken erforderlich sein, dann ist zu beachten, dass bei den ersten beiden Grafiken des 
Beispiels, die Grafiken ohne ODS GRAPHICS-Klammer, die Bezeichnung der Ordina-
tenachse gedreht wird. Bei der dritten Grafik wird die Beschriftung der Ordinatenachse 
nicht nur gedreht, sondern auch verkleinert. Des Weiteren werden die schräg gestellten 
Werte der Abszissenachse horizontal und verkleinert geschrieben.

